Здравствуйте.
Подскажите, пожалуйста, как сделать запрос, что бы получить в соответствую колонку Розничную цену, и в колонку рядом Закупочную?
Получить значение по одному типу цен не сложно, но как получить по двум типам?

```1c
    ТипЦенВыб = Справочники.ТипыЦенНоменклатуры.НайтиПоКоду("ЦБ000001");
    ТипЦенВыб1 = Справочники.ТипыЦенНоменклатуры.НайтиПоКоду("000000001");

    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    | ТоварыНаСкладахОстатки.Номенклатура.Ссылка КАК Ссылка,
    | ТоварыНаСкладахОстатки.Номенклатура КАК Номенклатура,
    | ТоварыНаСкладахОстатки.Номенклатура.Артикул КАК Артикул,
    | СУММА(ТоварыНаСкладахОстатки.КоличествоОстаток) КАК КоличествоОстаток,
    | ТоварыНаСкладахОстатки.Склад,
    | МестаХраненияНоменклатуры.МестоХранения,
    | ЦеныНоменклатурыСрезПоследних.Цена КАК ЦенаРозничная
    |ИЗ
    | РегистрНакопления.ТоварыНаСкладах.Остатки(&Дата, ) КАК ТоварыНаСкладахОстатки
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.МестаХраненияНоменклатуры КАК МестаХраненияНоменклатуры
    | ПО ТоварыНаСкладахОстатки.Номенклатура.Артикул = МестаХраненияНоменклатуры.Номенклатура.Артикул
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры.СрезПоследних(, ТипЦен = &ТипЦенВыб) КАК ЦеныНоменклатурыСрезПоследних
    | ПО ТоварыНаСкладахОстатки.Номенклатура.Артикул = ЦеныНоменклатурыСрезПоследних.Номенклатура.Артикул
    |ГДЕ
    | ТоварыНаСкладахОстатки.Номенклатура.Артикул В(&КОД)
    |
    |СГРУППИРОВАТЬ ПО
    | ТоварыНаСкладахОстатки.Номенклатура,
    | ТоварыНаСкладахОстатки.Номенклатура.Ссылка,
    | ТоварыНаСкладахОстатки.Номенклатура.Артикул,
    | ТоварыНаСкладахОстатки.Склад,
    | МестаХраненияНоменклатуры.МестоХранения,
    | ЦеныНоменклатурыСрезПоследних.Цена,
    | ЦеныНоменклатурыСрезПоследних.Цена";

    Запрос.УстановитьПараметр("КОД", МассивАртикула);
    Запрос.УстановитьПараметр("Дата", ТекущаяДата());
    Запрос.УстановитьПараметр("ТипЦенВыб", ТипЦенВыб);
    Запрос.УстановитьПараметр("ТипЦенВыб1", ТипЦенВыб1)
```
---
Под рукой нет подходящей конфигурации, но по идее вот так должно заработать

```1c
    "ВЫБРАТЬ
    | ТоварыНаСкладахОстатки.Номенклатура.Ссылка КАК Ссылка,
    | ТоварыНаСкладахОстатки.Номенклатура КАК Номенклатура,
    | ТоварыНаСкладахОстатки.Номенклатура.Артикул КАК Артикул,
    | СУММА(ТоварыНаСкладахОстатки.КоличествоОстаток) КАК КоличествоОстаток,
    | ТоварыНаСкладахОстатки.Склад,
    | МестаХраненияНоменклатуры.МестоХранения,
    | Розница.Цена КАК ЦенаРозничная,
    | Закупка.Цена КАК ЦенаЗакупочная
    |ИЗ
    | РегистрНакопления.ТоварыНаСкладах.Остатки(&Дата, ) КАК ТоварыНаСкладахОстатки
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.МестаХраненияНоменклатуры КАК МестаХраненияНоменклатуры
    | ПО ТоварыНаСкладахОстатки.Номенклатура = МестаХраненияНоменклатуры.Номенклатура
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры.СрезПоследних(, ТипЦен = &ТипЦенВыб) КАК Розница
    | ПО ТоварыНаСкладахОстатки.Номенклатура = Розница.Номенклатура
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры.СрезПоследних(, ТипЦен = &ТипЦенВыб1) КАК Закупка
    | ПО ТоварыНаСкладахОстатки.Номенклатура = Закупка.Номенклатура
    |ГДЕ
    | ТоварыНаСкладахОстатки.Номенклатура.Артикул В(&КОД)
    |
    |СГРУППИРОВАТЬ ПО
    | ТоварыНаСкладахОстатки.Номенклатура,
    | ТоварыНаСкладахОстатки.Номенклатура.Ссылка,
    | ТоварыНаСкладахОстатки.Номенклатура.Артикул,
    | ТоварыНаСкладахОстатки.Склад,
    | МестаХраненияНоменклатуры.МестоХранения,
    | Розница.Цена,
    | Закупка.Цена";
```
---
Добавьте еще один СрезПоследних по таблице Цен аналогично первому. Объединять ее надо тоже по Артикулу с ТоварыНаСкладахОстатки...что кстати в корне неправильно, объединять надо по ссылке (Номенклатуре), не смотря на то, что артикул судя по входящим данным будет уникальным, но тем не менее. Вот отбор по Артикулу ещё куда ни шло.
---
Я пробовал добавить ещё раз Цена из СрезПоследних, но не пойму как присвоить ему нужный параметр (другая цена). Выводит два раза Розничную цену.
---
, Естественно. Потому что вы берете цену из таблицы, в которой уже сделали отбор по "розничной цене".
Вам надо добавить ЕЩЁ одну таблицу СрезПоследних с отбором по "закупочной цене".

```1c
    |ИЗ
    | РегистрНакопления.ТоварыНаСкладах.Остатки(&Дата, ) КАК ТоварыНаСкладахОстатки
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.МестаХраненияНоменклатуры КАК МестаХраненияНоменклатуры
    | ПО ТоварыНаСкладахОстатки.Номенклатура = МестаХраненияНоменклатуры.Номенклатура
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры.СрезПоследних(, ТипЦен = &ТипЦенВыб) КАК ЦеныНоменклатурыСрезПоследних
    | ПО ТоварыНаСкладахОстатки.Номенклатура= ЦеныНоменклатурыСрезПоследних.Номенклатура
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры.СрезПоследних(, ТипЦен = &ТипЦенВыб1) КАК ЦеныНоменклатурыСрезПоследних1
    | ПО ТоварыНаСкладахОстатки.Номенклатура = ЦеныНоменклатурыСрезПоследних1.Номенклатура
```

Что-то вроде этого
---
Мой код, один в один) Он как раз и даёт на выходе одну и туже цену.
---
Приведите полный код Запроса на данный момент..
---
Пожалуйста.
```1c
    ТипЦенВыб = Справочники.ТипыЦенНоменклатуры.НайтиПоКоду("ЦБ000001");
    ТипЦенВыб1 = Справочники.ТипыЦенНоменклатуры.НайтиПоКоду("000000001");

    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    | ТоварыНаСкладахОстатки.Номенклатура.Ссылка КАК Ссылка,
    | ТоварыНаСкладахОстатки.Номенклатура КАК Номенклатура,
    | ТоварыНаСкладахОстатки.Номенклатура.Артикул КАК Артикул,
    | СУММА(ТоварыНаСкладахОстатки.КоличествоОстаток) КАК КоличествоОстаток,
    | ТоварыНаСкладахОстатки.Склад,
    | МестаХраненияНоменклатуры.МестоХранения,
    | ЦеныНоменклатурыСрезПоследних.Цена КАК ЦенаРозничная,
    | ЦеныНоменклатурыСрезПоследних.Цена КАК ЦенаЗакупочная
    |ИЗ
    | РегистрНакопления.ТоварыНаСкладах.Остатки(&Дата, ) КАК ТоварыНаСкладахОстатки
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.МестаХраненияНоменклатуры КАК МестаХраненияНоменклатуры
    | ПО ТоварыНаСкладахОстатки.Номенклатура.Артикул = МестаХраненияНоменклатуры.Номенклатура.Артикул
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры.СрезПоследних(, ТипЦен = &ТипЦенВыб) КАК ЦеныНоменклатурыСрезПоследних
    | ПО ТоварыНаСкладахОстатки.Номенклатура.Артикул = ЦеныНоменклатурыСрезПоследних.Номенклатура.Артикул
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры.СрезПоследних(, ТипЦен = &ТипЦенВыб1) КАК ЦеныНоменклатурыСрезПоследних1
    | ПО ТоварыНаСкладахОстатки.Номенклатура.Артикул = ЦеныНоменклатурыСрезПоследних1.Номенклатура.Артикул
    |ГДЕ
    | ТоварыНаСкладахОстатки.Номенклатура.Артикул В(&КОД)
    |
    |СГРУППИРОВАТЬ ПО
    | ТоварыНаСкладахОстатки.Номенклатура,
    | ТоварыНаСкладахОстатки.Номенклатура.Ссылка,
    | ТоварыНаСкладахОстатки.Номенклатура.Артикул,
    | ТоварыНаСкладахОстатки.Склад,
    | МестаХраненияНоменклатуры.МестоХранения,
    | ЦеныНоменклатурыСрезПоследних.Цена,
    | ЦеныНоменклатурыСрезПоследних.Цена";

    Запрос.УстановитьПараметр("КОД", МассивАртикула);
    Запрос.УстановитьПараметр("Дата", ТекущаяДата());
    Запрос.УстановитьПараметр("ТипЦенВыб", ТипЦенВыб);
    Запрос.УстановитьПараметр("ТипЦенВыб1", ТипЦенВыб1);
```
---
Под рукой нет подходящей конфигурации, но по идее вот так должно заработать

```1c
    "ВЫБРАТЬ
    | ТоварыНаСкладахОстатки.Номенклатура.Ссылка КАК Ссылка,
    | ТоварыНаСкладахОстатки.Номенклатура КАК Номенклатура,
    | ТоварыНаСкладахОстатки.Номенклатура.Артикул КАК Артикул,
    | СУММА(ТоварыНаСкладахОстатки.КоличествоОстаток) КАК КоличествоОстаток,
    | ТоварыНаСкладахОстатки.Склад,
    | МестаХраненияНоменклатуры.МестоХранения,
    | Розница.Цена КАК ЦенаРозничная,
    | Закупка.Цена КАК ЦенаЗакупочная
    |ИЗ
    | РегистрНакопления.ТоварыНаСкладах.Остатки(&Дата, ) КАК ТоварыНаСкладахОстатки
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.МестаХраненияНоменклатуры КАК МестаХраненияНоменклатуры
    | ПО ТоварыНаСкладахОстатки.Номенклатура = МестаХраненияНоменклатуры.Номенклатура
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры.СрезПоследних(, ТипЦен = &ТипЦенВыб) КАК Розница
    | ПО ТоварыНаСкладахОстатки.Номенклатура = Розница.Номенклатура
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры.СрезПоследних(, ТипЦен = &ТипЦенВыб1) КАК Закупка
    | ПО ТоварыНаСкладахОстатки.Номенклатура = Закупка.Номенклатура
    |ГДЕ
    | ТоварыНаСкладахОстатки.Номенклатура.Артикул В(&КОД)
    |
    |СГРУППИРОВАТЬ ПО
    | ТоварыНаСкладахОстатки.Номенклатура,
    | ТоварыНаСкладахОстатки.Номенклатура.Ссылка,
    | ТоварыНаСкладахОстатки.Номенклатура.Артикул,
    | ТоварыНаСкладахОстатки.Склад,
    | МестаХраненияНоменклатуры.МестоХранения,
    | Розница.Цена,
    | Закупка.Цена";
```
---
Да, вы правы, всё работает. Большое спасибо.
---
Пожалуйста)